

## Electronic Lock B 14/3/21

```
// This program uses bitRead and a shift register to display the numbers.  
// The code is entered as a three digit integer.  
// The digits are selected using the pot and the button.
```

```
byte lockServo = 8;  
byte Data = 7;  
byte Klock = 6;  
byte Latch = 5;  
byte button = 4;  
byte LOCKED = 3;  
byte UNLOCKED = 2;  
byte pot = A0;  
byte number;  
    // ZERO, ONE, TWO, THREE, FOUR, FIVE, SIX, SEVEN, EIGHT, NINE, blank  
byte NUMBERS[] {0x7E, 0x30, 0x6D, 0x79, 0x33, 0x5B, 0x5F, 0x70, 0x7F, 0x7B, 0x00};  
byte blank = 10;  
int potValue;  
int CODE = 274;    //This is the number to unlock.(user selectable.)  
int selNum;        //This is the number that has been selected.  
int LockedPos = 400;    //Servo position for Locked.  
int UnlockedPos = 1600; //Servo position for Unlocked.  
int pos;           //Timing for servo data on.  
int posneg;        //Timing for servo data off.  
int servoReps = 40; //Repetitions for servoDrive.
```

```
void setup() {  
    pinMode(Data, OUTPUT);  
    pinMode(Klock, OUTPUT);  
    pinMode(Latch, OUTPUT);  
    pinMode(lockServo, OUTPUT);  
    pinMode(LOCKED, OUTPUT);  
    pinMode(UNLOCKED, OUTPUT);  
    pinMode(button, INPUT);  
    pinMode(pot, INPUT);  
}  
  
void Klocking() {  
    digitalWrite(Klock, 1);  
    digitalWrite(Klock, 0);  
}  
  
void Latching() {  
    digitalWrite(Latch, 1);  
    digitalWrite(Latch, 0);  
}
```

```
void segDisp() {  
    for(int bitCount = 0; bitCount < 8; bitCount++) {  
        digitalWrite(Data, bitRead(NUMBERS[number], bitCount));  
        Klocking();  
    }  
    Latching();  
}
```

```
void readDial() {  
    while(digitalRead(button) == LOW) {  
        potValue = analogRead(pot);  
        number = potValue/103;    //1023/100 is 10, and we only need 9  
        segDisp();  
        delay(10);  
    }  
}
```



```

void servoDrive() {
    posneg = 2000 - pos;
    for(int reps = 0;reps <= servoReps;reps++){
        digitalWrite(lockServo,HIGH);
        delayMicroseconds(500);
        delayMicroseconds(pos);
        digitalWrite(lockServo,LOW);
        delayMicroseconds(posneg);
        delayMicroseconds(500);
        delay(17);
    }
}

void loop() {
    pos = LockedPos;
    servoDrive();
    digitalWrite(LOCKED,1);
    digitalWrite(UNLOCKED,0);
    number = blank;
    segDisp();
    while(digitalRead(button) == LOW){
        delay(10);
    }
    delay(500);
    readDial();
    selNum = number*100;           //Sets the first digit of the selNum.
    number = blank;
    segDisp();
    delay(500);
    readDial();
    selNum = selNum + (number*10); //Sets the second digit of the selNum.
    number = blank;
    segDisp();
    delay(500);
    readDial();
    selNum = selNum + number;    //Sets the third digit of the selNum.
    number = blank;
    segDisp();
    delay(500);
    if(selNum == CODE){         //Compares the selNum to the CODE.
        digitalWrite(LOCKED,0);
        digitalWrite(UNLOCKED,1);
        pos = UnlockedPos;
        servoDrive();
        while(digitalRead(button) == LOW){
            delay(10);
        }
    }
    else{
        for(int flashCount = 0;flashCount < 7;flashCount++){
            digitalWrite(LOCKED,0);
            delay(200);
            digitalWrite(LOCKED,1);
            delay(200);
        }
    }
    delay(1000);
    selNum = 0;
}

```

